How to Recognize Video Mode and Generate Free Running Synchronization Signals Using TSC8051C1/C2 Microcontrollers

Description

The TSC8051C1 is an application specific microcontroller for autosync monitor and digital control application. It includes the TEMIC static 8–bit 80C51 CPU core with 8 Kbytes of ROM and 256 bytes of RAM, 12x8–bit PWM channels, buffered HSYNC and VSYNC outputs, a watchdog timer and a multimaster I^2C controller.

This application note describes how to automatically recognize video mode by measuring the period and polarity of horizontal and vertical synchronization signals; it also explains how to generate free running synchronization signal for burn–in purpose.

In the rest of the application note, the use of words Hsync and Vsync means horizontal synchronization signal and vertical synchronization signal respectively.

Typical Autosync Monitor Application

The introduction of the TSC8051C1 in CRT monitors allows manufacturer and final user to get maximum flexibility.

- Automatic parameters adjustment during factory set–up.
- Auto–alignment capabilities.
- Saving of factory default parameters.
- Versatile frequency range up to 100KHz.
- More adjustment parameters are available to the user.
- Automatic video mode recognition that allows automatic monitor adjustment to the values previously saved by user.
- On chip I^2C bus controller allows Access bus implementation and so monitor adjustment by the PC's Keyboard.

Figure 1 shows a block diagram of a typical autosync monitor designed with the TSC8051C1.



Figure 1. Autosync monitor block diagram with the TSC8051C1

Hardware Description

TSC8051C1 implements some special features to allow video mode recognition without adding any external components.

- Special Hsync and Vsync inputs.
  - Vsync can generate an interrupt on either falling or rising edge. As 8051 core samples inputs one time per machine cycle, pulse duration less than Tosc x 12 (1µs using 12 MHz crystal) are not 100% detected. To allow Hsync pulses counting (duration>150ns), pulses are lengthened up to 1 cycle period to be sampled by the 8051 core. Figure 2 and Figure 3 show the VSYNC and HSYNC input block diagrams.
  - These features are programmable through EICON SFR (address E4h).

MSB **EICON SFR E4h** **LSB**

–	–	–	–	–	T0L	T0S	I0L

Symbol	Position	Name and Function
I0L	EICON.0	INT0/VSYNC input Level bit. Setting this bit inverts INT0/VSYNC input signal. Clearing it allows standard use of INT0/VSYNC input.
T0S	EICON.1	T0/HSYNC input Selection bit. Setting this bit allows short pulse capture. Clearing it allows standard use of T0/HSYNC input.
T0L	EICON.2	T0/HSYNC input Level bit. Setting this bit allows positive pulse capture. Clearing it allows negative pulse capture.

- Special Hsync and Vsync outputs.
- TSC8051C1 implements programmable Hsync and Vsync outputs. User can disable and can invert these outputs to provide good polarity to deflection stages.
- These features are programmable through SOCR SFR (address E5h).

MSB **SOCR SFR E5h** **LSB**

–	–	VOS	HOS	VOP	VOE	HOP	HOE

Symbol	Position	Name and Function
HOE	SOCR.0	HSYNC Output Enable bit. Setting this bit enables the HSYNC signal.
HOP	SOCR.1	HSYNC Output Polarity bit. Setting this bit inverts the HSYNC output.
VOE	SOCR.2	VSYNC Output Enable bit. Setting this bit enables the VSYNC signal.
VOP	SOCR.3	VSYNC Output Polarity bit. Setting this bit inverts the VSYNC output.
HOS	SOCR.4	HSYNC Output Selection bit. Setting this bit selects the VSYNC output, clearing it selects P3.5 SFR bit.
VOS	SOCR.5	VSYNC Output Selection bit. Setting this bit selects the VSYNC output, clearing it selects P3.3 SFR bit.



Figure 2. INT0/VSYNC input block diagram



Figure 3. T0/HSYNC input block diagram



Figure 4. HSYNC and VSYNC outputs block diagram

Video Mode Recognition Description

Vsync input is programmed to generate an interrupt each time a falling edge appears on Vsync.

Vsync period measurement, Vsync polarity detection and Hsync pulse counting are performed using Timer 0. Figure 5 shows the Timer 0 block diagram in mode 1.



Figure 5. Timer/Counter 0 in mode 1: 16–bit Counter

The measurement cycle is divided in 3 operations (3 Hsync frames):

Timer 0 is reset and Vsync interrupt is enabled.

- Vsync frequency measurement:
- In first interrupt of the cycle, timer 0 is programmed to be used as free running timer with fosc/12 clock. TR0 bit is set to start counting (GATE bit is reset). In the second interrupt, TR0 bit is reset to stop counting. At this time, TH0 and TL0 registers contain a representative value of the Vsync period (in µs if 12MHz crystal is used) (see Figure 6).



Figure 6. Vsync frequency measurement

- Vsync polarity detection:
- In the second interrupt of the cycle, timer 0 is programmed to be used as gated timer with fosc/12 clock. GATE bit is set and timer counts only during Vsync high level. In the third interrupt, TR0 bit is reset to stop counting. At this time, TH0 and TL0 registers contain a representative value of the Vsync high level duration (in µs if 12MHz crystal is used). If this duration is higher than the Vsync period divided by 2 then, Vsync has a negative polarity else it has a positive polarity (see Figure 7).



Figure 7. Vsync polarity detection

- Hsync polarity detection:
- In the second interrupt of the cycle, during the Vsync high level duration measurement, the Hsync signal is sampled 16 times, if number of high level samples is greater than 8 then, Hsync has a negative polarity else

it has a positive polarity (see Figure 8). Hsync input filter is then set to accept negative pulses or positive pulses respectively.



Figure 8. Hsync polarity detection

- Hsync frequency measurement:
- In the third interrupt of the cycle, timer 0 is programmed to be used as external event counter with Hsync clock. TR0 bit is set to start counting. In the fourth interrupt (last of the measurement cycle), TR0 bit is reset to stop counting. At this time, TH0 and TL0 registers contain a representative value of the Hsync period that is the number of Hsync pulses during a Vsync period (see Figure 9). A flag is set to inform main program of the end of cycle.



Figure 9. Hsync frequency measurement

When one cycle is completed, the main program checks the values and determines whether if the video mode has changed or not. If yes, the actions to take are listed hereafter:

- Depending on the Hsync frequency, S correction capacitors have to be updated.
- Some PWM values are updated.
- Video mute is activated.
- A research is made in EEPROM to find if the same video mode is already stored.
- If yes recall user set–up (update PWM values) from EEPROM, else default set–up is applied and the video mute is released.

Free Running Generation Description

During manufacturing burn–in, monitors are powered, but no video source is connected to the monitor. To force deflection stages' activity, free running Hsync and Vsync are output.

The software solution for free running generation, offers to the user a maximum of flexibility to program the best frequencies according to the deflection stages.

Two examples are proposed. In the first one, Vsync is a 60.1Hz negative polarity signal with 66µs pulses and Hsync is a 41.7KHz negative polarity signal with 2µs pulses, in the second one Vsync is a 72Hz positive polarity signal with 58µs pulses and Hsync is a 62.5KHz positive polarity signal with 1µs pulses.

Software Description

The software proposed hereafter is divided in two main routines:

- The Vsync Interrupt service routine.
- The Hsync/Vsync free running generation routine.

All the routines are based on a 12MHz oscillator operation; so 1 machine cycle has exactly 1µs duration. In the example, deflection stages are considered having a negative polarity synchronization input. User can program positive Hsync/Vsync outputs, by modifying the software as follows:

```
mov          A,#HOUT_VOUT_ENA
mov          C,Vpol_m
cpl          C    ; set VOP bit for positive
mov          ACC.3,C              ; polarity on Vout
mov          C,Hpol_m
cpl          C    ; set HOP bit for positive
mov          ACC.1,C              ; polarity on Hout
mov          SOCR,A               ; update Vout/Hout polarity
```

Vsync interrupt service routine has the highest priority. Due to the sampling clock, the Vsync period has a basic precision of 1µs. Depending on the instruction executed during the interrupt activation the measured period may be increased up to 4µs. The validation of a new detected video mode is effective only when the difference between the new measured and the previously saved period/counting is significant. This is achieved by the Check_diff subroutine.

The reception of a character on serial port activates one of the two Hsync/Vsync free running generation routine: '1' for the first example, '2' for the second one.

As the generation of the synchronization signals uses 100% of the CPU time, the only ways to disable generation are to clear the activation flag during an interrupt service routine (in the example, the flag is cleared in Vsync interrupt when a video source is input) or to apply a reset.

The listing includes the file reg51c1.inc that is the TSC8051C1 register declarations.

After a new video mode recognition, parameters are stored in the following variables:

```
Vpol_s      Vsync polarity
Hpol_s      Hsync polarity
Vperl_s     Vsync period high order byte
Vperh_s     Vsync period low order byte
Hcntl_s     Hsync count high order byte
Hcnth_s     Hsync count low order byte
```

The table hereafter presents different video modes and their associated parameters.

MODE	H. Frequency (KHz)	V. Frequency (Hz)	H. Polarity (Hpol_s)	V. Polarity (Vpol_s)	H. Count (Hcnt_s)	V.period (Vper_s)
EGA 640x350	31.5	70	+ (1)	– (0)	1C2h	37CEh
CGA 640x400	31.5	70	– (0)	+ (1)	1C2h	37CEh
VGA 640x480	31.5	60	– (0)	– (0)	20Dh	411Bh
VGA plus 800x600	35.5	56	+ (1)	– (0)	276h	45C1h
SVGA 800x600	37.8	60	+ (1)	+ (1)	276h	411Bh
VESA 800x600	48	72	+ (1)	+ (1)	29Ch	3641h
8514/a 1024x768	35.5	87	+ (1)	– (0)	198h	2CE6h
1280x1024	63.5	60	+ (1)	+ (1)	422h	411Bh

```
1                ; TEMIC 1996.
2                ; Demonstration program for video mode recognition
3                ; and free running generation with TSC 8051C1
4
5                $RB (0,1)                              ; bank 0 and 1 reserved
             6   $INCLUDE  (reg51c1.inc)  ; register declarations
             7   ; TEMIC 1996.
             8   ; Register declarations for TSC 8051C1 microcontroller
             9   ; Rev. A
            10
            11   ; BYTE Registers
0080        12   P0              DATA    080H
0090        13   P1              DATA    090H
00A0        14   P2              DATA    0A0H
00B0        15   P3              DATA    0B0H
            16
00D0        17   PSW             DATA    0D0H
00E0        18   ACC             DATA    0E0H
00F0        19   B               DATA    0F0H
0081        20   SP              DATA    081H
0082        21   DPL             DATA    082H
0083        22   DPH             DATA    083H
0087        23   PCON            DATA    087H
0088        24   TCON            DATA    088H
0089        25   TMOD            DATA    089H
008A        26   TL0             DATA    08AH
008B        27   TL1             DATA    08BH
008C        28   TH0             DATA    08CH
008D        29   TH1             DATA    08DH
00A8        30   IE              DATA    0A8H
00B8        31   IP              DATA    0B8H
0098        32   S0CON           DATA    098H
0099        33   S0BUF           DATA    099H
            34
00D8        35   S1CON           DATA    0D8H
00D9        36   S1STA           DATA    0D9H
00DA        37   S1DAT           DATA    0DAH
00DB        38   S1ADR           DATA    0DBH
            39
00AF        40   MSCON           DATA    0AFh
00E4        41   EICON           DATA    0E4h
00E5        42   SOCR            DATA    0E5h
00E6        43   HWDR            DATA    0E6h
00DF        44   PWMCON          DATA    0DFh
00E7        45   MXCRo           DATA    0E7h
00D7        46   MXCR1           DATA    0D7h
00EC        47   PWM0            DATA    0ECh
00ED        48   PWM1            DATA    0EDh
00EE        49   PWM2            DATA    0EEh
00EF        50   PWM3            DATA    0EFh
00F4        51   PWM4            DATA    0F4h
00F5        52   PWM5            DATA    0F5h
00F6        53   PWM6            DATA    0F6h
00F7        54   PWM7            DATA    0F7h
00FC        55   PWM8            DATA    0FCh
00FD        56   PWM9            DATA    0FDh
00FE        57   PWM10           DATA    0FEh
00FF        58   PWM11           DATA    0FFh
            59
            60
            61   ; BIT Registers
            62   ; PSW
00D7        63   CY              BIT     0D7H
```

```
00D6        64    AC              BIT     0D6H
00D5        65    F0              BIT     0D5H
00D4        66    RS1             BIT     0D4H
00D3        67    RS0             BIT     0D3H
00D2        68    OV              BIT     0D2H
00D0        69    P               BIT     0D0H
            70
            71    ; TCON
008F        72    TF1             BIT     08FH
008E        73    TR1             BIT     08EH
008D        74    TF0             BIT     08DH
008C        75    TR0             BIT     08CH
008B        76    IE1             BIT     08BH
008A        77    IT1             BIT     08AH
0089        78    IE0             BIT     089H
0088        79    IT0             BIT     088H
            80
            81    ; IE
00AF        82    EA              BIT     0AFH
00AD        83    ES1             BIT     0ADH
00AC        84    ES0             BIT     0ACH
00AB        85    ET1             BIT     0ABH
00AA        86    EX1             BIT     0AAH
00A9        87    ET0             BIT     0A9H
00A8        88    EX0             BIT     0A8H
            89
            90    ; IP0
00BD        91    PS1             BIT     0BDH
00BC        92    PS0             BIT     0BCH
00BB        93    PT1             BIT     0BBH
00BA        94    PX1             BIT     0BAH
00B9        95    PT0             BIT     0B9H
00B8        96    PX0             BIT     0B8H
            97
            98    ; P3
00B7        99    RD              BIT     0B7H
00B7        100   SDA             BIT     0B7H
00B6        101   WR              BIT     0B6H
00B6        102   SCL             BIT     0B6H
00B5        103   T1              BIT     0B5H
00B5        104   HOUT            BIT     0B5H
00B4        105   T0              BIT     0B4H
00B4        106   HSYNC           BIT     0B4H
00B3        107   INT1            BIT     0B3H
00B3        108   VOUT            BIT     0B3H
00B2        109   INT0            BIT     0B2H
00B2        110   VSYNC           BIT     0B2H
00B1        111   TXD             BIT     0B1H
00B0        112   RXD             BIT     0B0H
            113
            114   ; S0CON
009F        115   SM0             BIT     09FH
009E        116   SM1             BIT     09EH
009D        117   SM2             BIT     09DH
009C        118   REN             BIT     09CH
009B        119   TB8             BIT     09BH
009A        120   RB8             BIT     09AH
0099        121   TI              BIT     099H
0098        122   RI              BIT     098H
            123
            124   ; S1CON
00D85             CR0             BIT     0D8H
00D9        126   CR1             BIT     0D9H
```

```
00DA          127   AA              BIT     0DAH
00DB          128   SI              BIT     0DBH
00DC          129   STO             BIT     0DCH
00DD          130   STA             BIT     0DDH
00DE          131   ENS1            BIT     0DEH
              132
              133
              134
              135
              136   ; CONSTANT DEFINITION
              137   ; -------------------
0080          138   WDT_PER         EQU     80h             ; 2s watchdog period
              139
              140   HOUT_VOUT_SET   EQU     00101000b       ; Hout/Vout=1
00D7          141   HOUT_VOUT_CLR   EQU     11010111b       ; Hout/Vout=0
0035          142   HOUT_VOUT_ENA   EQU     00110101b       ; Hout/Vout enabled
00CF          143   HOUT_VOUT_DIS   EQU     11001111b       ; Hout/Vout disabled
              144
0002          145   EICON_H_NEG     EQU     00000010b       ; negative Hsync selection
0006          146   EICON_H_POS     EQU     00000110b       ; positive Hsync selection
              147
0008          148   VSYNC_DIFF      EQU     8               ; 7us=Vsync diff authorised
0002          149   HSYNC_DIFF      EQU     2               ; 1 pulse=Hsync diff authorised
              150
              151   ; BIT VARIABLE DEFINITION
              152   ; -----------------------
----          153                   BSEG AT 20h
0020          154   Vpol_m:         DBIT    1               ; measured Vsync polarity
0021          155   Hpol_m:         DBIT    1               ; measured Hsync polarity
0022          156   Vpol_s:         DBIT    1               ; saved Vsync polarity
0023          157   Hpol_s:         DBIT    1               ; saved Hsync polarity
0024          158   End_cycle:      DBIT    1               ; measuring end cycle flag (1)
0025          159   Free_run:       DBIT    1               ; Free running generation flag (1)
              160
              161   ; DATA VARIABLE DEFINITION
              162   ; ------------------------
----          163                   DSEG AT 30h
0030          164   Isr_state:      DS      1               ; Interrupt state flag
0031          165   Vperl_m:        DS      1               ; measured period high order byte
0032          166   Vperh_m:        DS      1               ; measured period low order byte
0033          167   Hcntl_m:        DS      1               ; measured count high order byte
0034          168   Hcnth_m:        DS      1               ; measured count low order byte
0035          169   Vperl_s:        DS      1               ; saved period high order byte
0036          170   Vperh_s:        DS      1               ; saved period low order byte
0037          171   Hcntl_s:        DS      1               ; saved count high order byte
0038          172   Hcnth_s:        DS      1               ; saved count low order byte
              173
0039          174   Stack:          DS      10h             ; 16 bytes stack
              175
              176
              177
              178   ;=======================================================================
              179   ;                       BEGIN CODE
              180   ;=======================================================================
              181
              182                   USING   0               ; RB0 used by default
----          183                   CSEG
0000          184                   ORG     0000h           ; reset address
0000 0106     185                   ajmp    Reset
              186
0003          187                   ORG     0003h           ; VSYNC (INT0) interrupt
0003 0200FC   188                   ljmp    Vsync_isr
              189
```

```
                  190
                  191   ;=======================================================================
                  192   ;                         INITIALISATION
                  193   ;=======================================================================
                  194
 0006 758138      195   Reset:           mov     SP,#Stack-1   ; stack pointer initialisation
                  196
 0009 78FF        197                    mov     R0,#0FFh      ; Internal RAM initialisation
 000B 7600        198   Ram_init:        mov     @R0,#00H
 000D D8FC        199                    djnz    R0,Ram_init
                  200
 000F 758DE6      201                    mov     TH1,#0E6h     ; T1 used as baud rate generator
 0012 758B00      202                    mov     TL1,#00h      ; at 1200 bauds with 12MHz crystal
 0015 758921      203                    mov     TMOD,#21h     ; T0 16b counter, T1 8b autoreload
 0018 758841      204                    mov     TCON,#41h     ; T1 run, INT0 falling edge
                  205
 001B 758700      206                    mov     PCON,#00h     ; SMOD=0
 001E 759852      207                    mov     S0CON,#52h    ; 8-bit UART, Rx enabled
                  208
 0021 75A881      209                    mov     IE,#81h       ; IE0 enabled
 0024 75B801      210                    mov     IP,#01h       ; IE0 high priority
                  211
 0027 75E680      212                    mov     HWDR,#WDT_PER ; watchdog activation
                  213
                  214
                  215   ;=======================================================================
                  216   ;                         MAIN PROGRAM
                  217   ;=======================================================================
                  218
 002A 202410      219   Wait_sync:       jb      End_cycle,Check_mode
 002D 75E680      220                    mov     HWDR,#WDT_PER ; watchdog refresh
                  221
                  222                    ; here must be inserted the
                  223                    ; man-machine interface control
                  224
 0030 3098F7      225                    jnb     RI,Wait_sync  ; example for free running
 0033 C298        226                    clr     RI
 0035 E599        227                    mov     A,S0BUF
                  228
 0037 B43104      229                    cjne    A,#'1',test_car
 003A 11B3        230                    acall   H_V_sync_gen_1; Free running generation
 003C 012A        231                    ajmp    Wait_sync
                  232
 003E B432E9      233   test_car:        cjne    A,#'2',Wait_sync
 0041 3104        234                    acall   H_V_sync_gen_2
                                                                ; Free running generation
 0043 012A        235                    ajmp    Wait_sync
                  236
 0045 E4          237   Check_mode:      clr     A
 0046 A220        238                    mov     C,Vpol_m
 0048 92E0        239                    mov     ACC.0,C
 004A A222        240                    mov     C,Vpol_s
 004C 9400        241                    subb    A,#00
 004E 702A        242                    jnz     Mode_changed  ; Vsync polarity changed
                  243
 0050 A221        244                    mov     C,Hpol_m
 0052 92E0        245                    mov     ACC.0,C
 0054 A223        246                    mov     C,Hpol_s
 0056 9400        247                    subb    A,#00
 0058 7020        248                    jnz     Mode_changed  ; Hsync polarity changed
                  249
 005A A831        250                    mov     R0,Vperl_m
 005C A932        251                    mov     R1,Vperh_m
```

```
 005E AA35    252                  mov     R2,Vperl_s
 0060 AB36    253                  mov     R3,Vperh_s
 0062 7C08    254                  mov     R4,#VSYNC_DIFF
 0064 11A0    255                  acall   Check_diff    ; compare new & old period
 0066 7012    256                  jnz     Mode_changed  ; Vsync period changed
              257
 0068 A833    258                  mov     R0,Hcntl_m
 006A A934    259                  mov     R1,Hcnth_m
 006C AA37    260                  mov     R2,Hcntl_s
 006E AB38    261                  mov     R3,Hcnth_s
 0070 7C02    262                  mov     R4,#HSYNC_DIFF
 0072 11A0    263                  acall   Check_diff    ; compare new & old counting
 0074 7004    264                  jnz     Mode_changed  ; Vsync period changed
              265
 0076 C224    266                  clr     End_cycle     ; a new cycle can start
 0078 012A    267                  ajmp    Wait_sync
              268
 007A         269  Mode_changed:
              270
              271  ; user define   setb    Video_mute    ; video mute during mode change
              272  ; user define   acall   Cs_select     ; S correction capacitors update
              273                                        ; depending on Vsync period
              274
 007A 7435    275                  mov     A,#HOUT_VOUT_ENA
 007C A220    276                  mov     C,Vpol_m      ; set VOP bit for negative
 007E 92E3    277                  mov     ACC.3,C       ; polarity on Vout
 0080 A221    278                  mov     C,Hpol_m      ; set HOP bit for negative
 0082 92E1    279                  mov     ACC.1,C       ; polarity on Hout
0084 F5E5     280                  mov     SOCR,A        ; update Vout/Hout polarity
              281
 0086 A220    282                  mov     C,Vpol_m
 0088 9222    283                  mov     Vpol_s,C      ; save new Vsync polarity
 008A 853135  284                  mov     Vperl_s,Vperl_m
 008D 853236  285                  mov     Vperh_s,Vperh_m
                                                         ; save new Vsync period
              286
 0090 A221    287                  mov     C,Hpol_m
 0092 9223    288                  mov     Hpol_s,C      ; save new Hsync polarity
 0094 853337  289                  mov     Hcntl_s,Hcntl_m
 0097 853438  290                  mov     Hcnth_s,Hcnth_m
                                                         ; save new Hsync period
              291
              292  ;    Here must be inserted the research of this video mode in EEPROM
              293  ;    if it is already stored, then recall user's screen parameters
              294  ;    if not, recall factory default screen parameters.
              295  ;    This new mode will be stored in EEPROM after user adjustments
              296
 009A 31F9    297                  acall   Out_results   ; send results to serial port
              298
              299  ; user define   clr     Video_mute    ; end of video mute
 009C C224    300                  clr     End_cycle     ; a new cycle can start
 009E 012A    301                  ajmp    Wait_sync
              302
              303
              304  ;=========================================================================
              305  ;                         SUBROUTINES
              306  ;=========================================================================
              307
              308  ;-------------------------------------------------------------------------
              309  ;
              310  ;      This subroutine checks if the absolute difference of two words
              311  ;      is less than a given byte value
              312  ;
```

```
                313  ;      Inputs:   R0:      word 1 low order byte
                314  ;                R1:      word 1 high order byte
                315  ;                R2:      word 2 low order byte
                316  ;                R3:      word 2 high order byte
                317  ;                R4:      limit of difference
                318  ;
                319  ;      Outputs:  A:       if A = 0 the difference is less than the limit
                320  ;                         else the difference is greater than or equal to
                                                the limit
                321  ;-------------------------------------------------------------------------
                322
00A0 C3         323  Check_diff:      clr      C
00A1 E9         324                   mov      A,R1
00A2 9B         325                   subb     A,R3          ; A=MSB difference
00A3 700D       326                   jnz      End_check     ; MSB not equal
                327
00A5 E8         328                   mov      A,R0
00A6 9A         329                   subb     A,R2          ; A=LSB difference
00A7 5003       330                   jnc      Check_pos
                331
00A9 C3         332                   clr      C             ; negative difference
00AA EA         333                   mov      A,R2
00AB 98         334                   subb     A,R0          ; A=LSB difference
                335
00AC B50400     336  Check_pos:       cjne     A,AR4,$+3
00AF 5001       337                   jnc      End_check
00B1 E4         338                   clr      A
00B2            339  End_check:       ret
                340
                341
                342
                343  ;-------------------------------------------------------------------------
                344  ;
                345  ;     This subroutine generates free Running synchronization Signals
                346  ;
                347  ;     Hout = 41.7KHz with 2us negative pulses
                348  ;     Vout = 60.1Hz with 66us negative pulses
                349  ;
                350  ;-------------------------------------------------------------------------
                351
00B3 E5E5       352  H_V_sync_gen_1:  mov      A,SOCR
00B5 54CF       353                   anl      A,#HOUT_VOUT_DIS
00B7 F5E5       354                   mov      SOCR,A        ; select P3.3/P3.5 as Vout/Hout
00B9 D225       355                   setb     Free_run      ; set flag (cleared in Vsync isr)
                356
00BB 53B0D7     357  V_pulse_1:       anl      P3,#HOUT_VOUT_CLR
00BE 00         358                   nop
00BF D2B5       359                   setb     HOUT          ; 2us neg pulse on Hout
00C1 790A       360                   mov      R1,#10
00C3 D9FE       361                   djnz     R1,$          ; 20us tempo
                362
00C5 C2B5       363  clr              HOUT
00C7 00         364                   nop
00C8 D2B5       365                   setb     HOUT          ; 2us neg pulse on Hout
00CA 790A       366                   mov      R1,#10
00CC D9FE       367                   djnz     R1,$          ; 20us tempo
                368
00CE C2B5       369                   clr      HOUT
00D0 00         370                   nop
00D1 D2B5       371                   setb     HOUT          ; 2us neg pulse on Hout
00D3 7907       372                   mov      R1,#7
00D5 D9FE       373                   djnz     R1,$          ; 14us tempo
00D7 D2B3       374                   setb     VOUT          ; 66us neg pulse on Vout
```

```
                375
00D9 78E6       376                     mov     R0,#230         ; 230 * 3 Hsync pulses
00DB 00         377   H_pulse_1:        nop
00DC 00         378                     nop
00DD 75E680     379                     mov     HWDR,#WDT_PER   ; watchdog refresh
                380
00E0 C2B5       381                     clr     HOUT
00E2 00         382                     nop
00E3 D2B5       383                     setb    HOUT            ; 2us neg pulse on Hout
00E5 790A       384                     mov     R1,#10
00E7 D9FE       385                     djnz    R1,$            ; 20us tempo
                386
00E9 C2B5       387                     clr     HOUT
00EB 00         388                     nop
00EC D2B5       389                     setb    HOUT            ; 2us neg pulse on Hout
00EE 790A       390                     mov     R1,#10
00F0 D9FE       391                     djnz    R1,$            ; 20us tempo
                392
00F2 C2B5       393                     clr     HOUT
00F4 00         394                     nop
00F5 D2B5       395                     setb    HOUT            ; 2us neg pulse on Hout
00F7 7906       396                     mov     R1,#6
00F9 D9FE       397                     djnz    R1,$            ; 12us tempo
                398
00FB 302505     399                     jnb     Free_run,End_gen_1
                                                                ; Free running enabled ?
                400
00FE D8DB       401                     djnz    R0,H_pulse_1
0100 00         402                     nop
0101 01BB       403                     ajmp    V_pulse_1
0103 22         404   End_gen_1:        ret
                405
                406
                407;-------------------------------------------------------------------------
                408   ;
                409   ;       This subroutine generates free Running synchronization Signals
                410   ;
                411   ;       Hout = 62.5KHz with 1us positive pulses
                412   ;       Vout = 72Hz with 58us positive pulses
                413   ;
                414;-------------------------------------------------------------------------
                415
0104 E5E5       416   H_V_sync_gen_2:   mov     A,SOCR
0106 54CF       417                     anl     A,#HOUT_VOUT_DIS
0108 F5E5       418                     mov     SOCR,A          ; select P3.3/P3.5 as Vout/Hout
010A D225       419                     setb    Free_run        ; set flag (cleared in Vsync isr)
                420
010C 53B0D7     421   V_pulse_2:        orl     P3,#HOUT_VOUT_SET
010F D2B5       422                     clr     HOUT            ; 1us pos pulse on Hout
0111 00         423                     nop
0112 7906       424                     mov     R1,#6
0114 D9FE       425                     djnz    R1,$            ; 12us tempo
                426
0116 C2B5       427                     setb    HOUT
0118 D2B5       428                     clr     HOUT            ; 1us pos pulse on Hout
011A 00         429                     nop
011B 7906       430                     mov     R1,#6
011D D9FE       431                     djnz    R1,$            ; 12us tempo
                432
011F C2B5       433                     setb    HOUT
0121 D2B5       434                     clr     HOUT            ; 1us pos pulse on Hout
0123 00         435                     nop
0124 7906       436                     mov     R1,#6
```

```
 0126 D9FE    437                  djnz    R1,$          ; 12us tempo
              438
 0128 C2B5    439                  setb    HOUT
 012A D2B5    440                  clr     HOUT          ; 1us pos pulse on Hout
 012C 00      441                  nop
 012D 7903    442                  mov     R1,#3
 012F D9FE    443                  djnz    R1,$          ; 6us tempo
 0131 D2B3    444                  clr     VOUT          ; 58us pos pulse on Vout
              445
 0133 78D8    446                  mov     R0,#216       ; 216 * 4 Hsync pulses
 0135 00      447  H_pulse_2:      nop
 0136 00      448                  nop
 0137 75E680  449                  mov     HWDR,#WDT_PER ;watchdog refresh
              450
 013A C2B5    451                  setb    HOUT
 013C D2B5    452                  clr     HOUT          ; 1us pos pulse on Hout
 013E 00      453                  nop
 013F 7906    454                  mov     R1,#6
 0141 D9FE    455                  djnz    R1,$          ; 12us tempo
              456
 0143 C2B5    457                  setb    HOUT
 0145 D2B5    458                  clr     HOUT          ; 1us pos pulse on Hout
 0147 00      459                  nop
 0148 7906    460                  mov     R1,#6
 014A D9FE    461                  djnz    R1,$          ; 12us tempo
              462
 014C C2B5    463                  setb    HOUT
 014E D2B5    464                  clr     HOUT          ; 1us pos pulse on Hout
 0150 00      465                  nop
 0151 7906    466                  mov     R1,#6
 0153 D9FE    467                  djnz    R1,$          ; 12us tempo
              468
 0155 C2B5    469                  setb    HOUT
 0157 D2B5    470                  clr     HOUT          ; 1us pos pulse on Hout
 0159 00      471                  nop
 015A 7902    472                  mov     R1,#2
 015C D9FE    473                  djnz    R1,$          ; 4us tempo
              474
 015E 302505  475                  jnb     Free_run,End_gen_2
                                                        ; Free running enabled ?
              476
 0161 D8D2    477                  djnz    R0,H_pulse_2
 0163 00      478                  nop
 0164 210C    479                  ajmp    V_pulse_2
 0166 22      480  End_gen_2:      ret
              481
              482
              483
              484
              485;-------------------------------------------------------------------------
              486  ;
              487  ;       This routine is the VSYNC interrupt service routine
              488  ;
              489;-------------------------------------------------------------------------
              490                  USING   1             ; RB1 used in interrupt
              491
0167 C225     492  Vsync_isr:      clr     Free_run      ; stop Free running
0169 302401   493                  jnb     End_cycle,Vsync_beg_isr
                                                        ; Cycle not yet treated
 016C 32      494                  reti
              495
 016D C0E0    496  Vsync_beg_isr: push    ACC
 016F C0D0    497  push            PSW
```

```
0171 75D008  498                 mov     PSW,#08h       ; RB1 selection
0174 E530    499                 mov     A,Isr_state    ; load ISR state
             500
0176 B4000F  501   State_0:       cjne    A,#00,state_1
0179 758921  502                 mov     TMOD,#21h      ; T0 : free running 16-bit timer
017C 758A00  503                 mov     TL0,#00h
017F 758C00  504                 mov     TH0,#00h
0182 D28C    505                 setb    TR0            ; start measuring Vsync period
0184 0530    506                 inc     Isr_state
0186 21F3    507                 ajmp    Vsync_end_isr
             508
0188 B40133  509   State_1:       cjne    A,#01,state_2
018B 00      510                 nop                    ; nop are inserted for timing
018C 00      511                 nop                    ; compensation between start
018D 00      512                 nop                    ; and stop counting
018E 00      513                 nop
018F C28C    514                 clr     TR0            ; stop T0
0191 858A31  515                 mov     Vperl_m,TL0
0194 858C32  516                 mov     Vperh_m,TH0    ; store Vsync period
0197 758929  517                 mov     TMOD,#29h      ; T0 : gated 16-bit timer
019A 758A00  518                 mov     TL0,#00h
019D 758C00  519                 mov     TH0,#00h
01A0 D28C    520                 setb    TR0            ; start measuring Vsync high level
             521
01A2 E4      522                 clr     A              ; Hsync polarity detection
01A3 7810    523                 mov     R0,#16         ; 16 samples
01A5 A2B4    524   S1_1:         mov     C,HSYNC        ; read pin level
01A7 3400    525                 addc    A,#00          ; store state
01A9 D8FA    526                 djnz    R0,S1_1
01AB B40800  527                 cjne    A,#08,S1_2
01AE 9221    528   S1_2:         mov     Hpol_m,C       ; store Hsync polarity (0 = neg)
01B0 4005    529                 jc      S1_3
01B2 75E402  530                 mov     EICON,#EICON_H_NEG
                                                        ; negative Hsync pulses selection
01B5 21BA    531                 ajmp    S1_4
01B7 75E406  532   S1_3:         mov     EICON,#EICON_H_POS
                                                        ; positive Hsync pulses selection
             533
01BA 0530    534   S1_4:         inc     Isr_state
01BC 21F3    535                 ajmp    Vsync_end_isr
             536
01BE B4021B  537   State_2:       cjne    A,#02,state_3
01C1 758925  538                 mov     TMOD,#25h      ; T0 : 16-bit counter
01C4 758A00  539                 mov     TL0,#00h
01C7 A88C    540                 mov     R0,TH0         ; save MSB
01C9 758C00  541                 mov     TH0,#00h
01CC D28C    542                 setb    TR0            ; start counting Hsync pulses
             543
01CE E532    544                 mov     A,Vperh_m      ; Vsync polarity detection
01D0 C3      545                 clr     C
01D1 13      546                 rrc     A              ; A = Vsync period / 20
01D2 B50800  547                 cjne    A,AR0,S2_1     ; test only high order byte
01D5 B3      548   S2_1:         cpl     C
01D6 9220    549                 mov     Vpol_m,C       ; store Vsync polarity (0 = neg)
01D8 0530    550                 inc     Isr_state
01DA 21F3    551                 ajmp    Vsync_end_isr
             552
01DC 00      553   state_3:      nop                    ; nop are inserted for timing
01DD 00      554                 nop                    ; compensation between start
01DE 00      555                 nop                    ; and stop counting
01DF 00      556                 nop
01E0 00      557                 nop
01E1 00      558                 nop
```

```
 01E2 00      559                    nop
 01E3 00      560                    nop
 01E4 00      561                    nop
 01E5 00      562                    nop
 01E6 C28C    563                    clr     TR0             ; stop T0
 01E8 858A33  564                    mov     Hcntl_m,TL0
 01EB 858C34  565                    mov     Hcnth_m,TH0     ; store Hsync pulse count
 01EE 753000  566                    mov     Isr_state,#00   ; reset ISR state to start a new
                                                              cycle
 01F1 D224    567                    setb    End_cycle       ; set flag for the main program
              568
 01F3 D0D0    569   Vsync_end_isr:   pop     PSW
 01F5 D0E0    570                    pop     ACC
 01F7 32      571                    reti
              572
              573                    USING   0               ; RB0 used by default
              574
              575
              576   ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
              577   ;                                                                     +
              578   ;                The subroutines hereafter are only used for
                                     demonstration program+
              579   ;                                                                     +
              580   ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
              581
              582   ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
              583   ; Send measured parameters on serial port                            +
              584   ; Vsync period, polarity                                             +
              585   ; Hsync counting, polarity                                           +
              586   ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
 01F8 900256  587   Out_results:     mov     DPTR,#Vsync_msge
 01FB 512F    588                    acall   Out_msge        ; display Vsync message
 01FD E536    589                    mov     A,Vperh_s
 01FF 5139    590                    acall   Out_byte
 0201 E535    591                    mov     A,Vperl_s
 0203 5139    592                    acall   Out_byte        ; display Vsync period
 0205 7420    593                    mov     A,#' '
 0207 514E    594                    acall   Out_char
 0209 742D    595                    mov     A,#'-'
 020B A222    596                    mov     c,Vpol_s
 020D 5002    597                    jnc     Vpol_neg
 020F 742B    598                    mov     A,#'+'
 0211 514E    599   Vpol_neg:        acall   Out_char        ; display Vsync polarity
              600
0213 90025F   601                    mov     DPTR,#Hsync_msge
0216 512F     602                    acall   Out_msge        ; display Hsync message
0218 E538     603                    mov     A,Hcnth_s
021A 5139     604                    acall   Out_byte
021C E537     605                    mov     A,Hcntl_s
021E 5139     606                    acall   Out_byte        ; display Hsync count
0220 7420     607                    mov     A,#' '
0222 514E     608                    acall   Out_char
0224 742D     609                    mov     A,#'-'
0226 A223     610                    mov     c,Hpol_s
0228 5002     611                    jnc     Hpol_neg
022A 742B     612                    mov     A,#'+'
022C 514E     613   Hpol_neg:        acall   Out_char        ; display Hsync polarity
022E 22       614                    ret
              615
              616   ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
              617   ; Send a message on serial port                                      +
              618   ; DPTR is the message address                                        +
              619   ; a NUL character is the end of message                              +
```

```
                620  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
 022F E4        621  Out_msge:        clr     A
 0230 93        622                  movc    A,@A+DPTR
 0231 6005      623                  jz      Out_end       ; last character
 0233 A3        624                  inc     DPTR
 0234 514E      625                  acall   Out_char
 0236 412F      626                  ajmp    Out_msge
 0238 22        627  Out_end:        ret
                628
                629  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
                630  ; Send a hexadecimal byte on serial port                              +
                631  ; A is the byte to send                                               +
                632  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
 0239 C0E0      633  Out_byte:        PUSH    ACC
 023B C4        634                  SWAP    A
 023C 5143      635                  ACALL   Out_nib
 023E D0E0      636                  POP     ACC
 0240 5143      637                  ACALL   Out_nib
 0242 22        638                  ret
                639
                640  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
                641  ; Send a hexadecimal nibble on serial port                            +
                642  ; A is the nibble to send                                             +
                643  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
 0243 540F      644  Out_nib:         ANL     A,#0FH
 0245 2490      645                  ADD     A,#90H
 0247 D4        646                  DA      A
 0248 3440      647                  ADDC    A,#40H
 024A D4        648                  DA      A
 024B 514E      649                  acall   Out_char
 024D 22        650                  ret
                651
                652  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
                653  ;Send an ASCII character on serial port                               +
                654  ; A is the character to send                                          +
                655  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
 024E 3099FD    656  Out_char:        jnb     TI,Out_char
 0251 C299      657                  clr     TI
 0253 F599      658                  mov     S0BUF,A
 0255 22        659                  ret
                660
                661  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
                662  ; Messages definition                                                 +
                663  ;+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
0256 0A0D5673   664  Vsync_msge:      DB      0Ah,0Dh,'Vsync ',0
 025A 796E6320
 025E 00
025F 0A0D4873   665  Hsync_msge:      DB      0Ah,0Dh,'Hsync ',0
 0263 796E6320
 0267 00
```